
Context


A context is a scope. It is core to any application or module built using Robotlegs.



Creating a Context


To create a context simply instantiate a new Context and provide some configuration:


_context = new Context()
    .install(MVCSBundle)
    .configure(
        MyModuleConfig,
        new ContextView(view));



Note: you must hold on to that context reference. Failing to do so will result in the context instance being garbage collected.


In the example above we are installing a bundle, a configuration and a reference to a Display Object Container. The Display Object Container will be used as the “contextView”.


The “contextView” should always be provided as the final configuration as it may trigger context initialization.







Bundles and Extensions


An extension integrates 3rd party code into a context.


A bundle groups a selection of extensions together into a single drop-in package.


Most extensions and bundles can be installed as classes:


_context.install(MVCSBundle, InjectorLoggingExtension);



Some extensions offer some extra configuration by way of constructor arguments:


_context.install(new EventDispatcherExtension(myExistingEventDispatcher));



Usually these extensions provide sensible default constructor arguments and can be installed as classes:


_context.install(EventDispatcherExtension);



Extensions [https://github.com/robotlegs/robotlegs-framework/tree/master/src/robotlegs/bender/extensions]





Configuration


Configurations can be provided as class references or object instances.



Class Configs


Configs are normally supplied as class references:


_context.configure(MyModuleConfig);



If the context has already been initialized an instance of the supplied class will be instantiated using the context’s injector.


If the context is not yet initialized the class will be queued. When the context is initialized an instance of the class will be instantiated.


Such a config might look like this:


class MyModuleConfig implements IConfig
{
    [Inject]
    public var mediatorMap:IMediatorMap;

    public function configure():void
    {
        mediatorMap.map(SomeView)
            .toMediator(SomeMediator);
    }
}



If a config implements the IConfig interface configure() will be invoked after construction/injection.





Object Configs


_context.configure(new MyModuleConfig("hello"));



If the context has already been initialized the supplied object will simply be injected into.


If the context is not yet initialized the object will be queued. When the context is initialized the object will be injected into.

Message Dispatcher



Listeners listen, handlers handle


The Event Dispatcher gives us a great way to observe objects, but the dispatch process is synchronous and listeners do not have the opportunity to suspend or terminate the dispatch. That’s completely fine when we’re just watching.


A Message Dispatcher is like an Event Dispatcher, but the dispatch has the potential to be asynchronous and handlers have the opportunity to suspend or terminate the dispatch.





Important


Message Dispatchers do not replace Event Dispatchers in any way. They solve completely different problems. The differences are explored in greater detail below.







Comparison



Overview


Event Dispatchers dispatch “events” to “listeners”. The dispatch is synchronous (blocking).


Message Dispatchers route “messages” through “handlers”. The dispatch may be synchronous or asynchronous depending on the handlers.


const user:User = new User("Sally");

const asyncHandler:Function = function(user:User, callback:Function):void {
  trace('Just gonna chill for, like, 1000 milliseconds or something...');
  setTimeout(callback, 1000);
};

const syncHandler:Function = function(user:User):void {
  trace('Golly, I synchronously observed: ' + user);
};

messageDispatcher.addMessageHandler(user, asyncHandler);
messageDispatcher.addMessageHandler(user, syncHandler);

messageDispatcher.dispatchMessage(user, function():void {
  trace('Dispatch eventually completed');
});






Events & Messages



Events (Event Dispatcher)


Events are objects of type Event:


new Event("hello")
new MouseEvent(MouseEvent.CLICK)






Messages (Message Dispatcher)


Messages are objects of any type:


"hello"
new User()








Listeners & Handlers



Listeners (Event Dispatcher)


Listeners are functions that accept objects of type Event:


function onHello(event:Event):void
function onClick(event:MouseEvent):void






Handlers (Message Dispatcher)


Handlers are functions that handle messages, optionally suspending, resuming or terminating the sequential dispatch.


function handleGreeting(greeting:String):void
function handleUser(user:User):void



A handler must have one of the following signatures:


function():void
function(message:Object):void
function(message:Object, callback:Function):void



Handlers can accept typed messages:


function handleGreeting(greeting:String):void
function handleUser(user:User):void



Handlers that accepts callbacks have the potential to be asynchronous:


function handleUser(user:User, callback:Function):void {
  setTimeout(callback, 1000);
}



Asynchronous handlers can terminate the dispatch by sending an error to the callback:


function handleUser(user:User, callback:Function):void {
  callback(new Error("something went wrong"));
}



Note: The error object can be of any type although sending actual Error instances is highly recommended. The following is not recommended:


function handleUser(user:User, callback:Function):void {
  callback("something went wrong");
}








Adding Listeners & Handlers



Adding Listeners (Event Dispatcher)


Listeners are added to event dispatchers by string keys:


eventDispatcher.addEventListener("hello", onHello)
eventDispatcher.addEventListener(MouseEvent.CLICK, onClick)






Adding Handlers (Message Dispatcher)


Handlers are added to message dispatchers with the message itself as the key:


messageDispatcher.addMessageHandler("hello", handleGreeting)
messageDispatcher.addMessageHandler(user, handleUser)








Dispatching Events & Messages



Dispatching an Event (Event Dispatcher)


Dispatching an event is synchronous (and immutable? write a test):


eventDispatcher.dispatchEvent(new Event("hello"));
// we can be sure that the dispatch has completed at this point






Dispatching a Message (Message Dispatcher)


Dispatching a message may be asynchronous (non-blocking), but the dispatch is always run in series.


messageDispatcher.dispatchMessage("hello");
// we can not be sure that the dispatch has completed at this point



Note: Dispatch might sometimes be synchronous, depending on the handlers.





Callbacks


Dispatch completion triggers a callback (optional):


messageDispatcher.dispatchMessage("hello", function():void {
  // dispatch has completed
});



A callback must have one of the following signatures:


function callback():void;
function callback(error:Object):void;






Errors


An error returned from a handler is sent to the original callback:


messageDispatcher.dispatchMessage(user, function(error:Object):void {
  if (error) {
    trace('dang, there was a problem: ' + error);
    throw error;
  } else {
    trace('Dispatch eventually completed');
  }
});



Note: One should always accept and handle errors in a callback.









Handler and Callback Quick Reference



Valid Handler Signatures


function():void
function(message:Object):void
function(message:Object, callback:Function):void






Valid Callback Signatures


function():void
function(error:Object):void
function(error:Object, message:Object):void








Considerations



Stack Depth


Sequential synchronous callbacks increase the stack depth. Consider the following handler:


function(message:Class, callback:Function):void {
    callback();
}



It should be clear that such a handler will increase the stack depth. If all handlers in a given dispatch follow that form there will eventually be a stack overflow. On my machine the overflow occurs after roughly 1400 synchronous handlers. However, as soon as an actual asynchronous handler is encountered the stack will drop back to normal as the dispatch will be spread over more than one frame:


function(message:Class, callback:Function):void {
    setTimeout(callback, 10);
}






Asynchronous Error Handling


Consider the following:


try
{
    dispatcher.dispatchMessage(Message);
}
catch(error:Error)
{
    trace(error);
}



It is important to understand that an error thrown by an asynchronous handler will not be caught by this try-catch as it will occur in a future turn of the event loop.


This is why it’s important that error handling be core to the asynchronous callback conventions employed by the framework:


dispatcher.dispatchMessage(Message, function(error:Error):void {
    if (error) throw error;
    trace("Completed successfully!");
});






Speed


A Message Dispatcher performs roughly the same as an Event Dispatcher for non-callback handlers. Handlers that accept callbacks slow things down and have the potential to increase the stack depth.







Background


The majority of the API is modeled on Event Dispatcher. The callback style is inspired by work going on in other evented environments where asynchronous control flow is important. See:



		http://nodeguide.com/style.html#callbacks


		https://github.com/caolan/async


		http://en.wikipedia.org/wiki/Futures_and_promises


		http://wiki.commonjs.org/wiki/Promises/A


		http://taskjs.org/


		http://api.jquery.com/category/deferred-object/


		https://github.com/briancavalier/when.js





The Message Dispatcher provides a simple alternative to a full blown Deferred, Promise or Future implementation, but at the cost of flexibility:


“Using a callback in a function is the poor man’s version of Promises. You can’t easily chain extra callbacks. However, it’s cheaper, it doesn’t require creating the extra promise objects and callback chain. Outward facing APIs should provide promises rather than callbacks. Internally the functions implemented with c++ just use a callback.” see: http://groups.google.com/group/nodejs/msg/10ad8f49535910bf





Dev Notes


The dispatch chain is presently frozen at the time of dispatch. Explore options for manipulation of the dispatch chain mid-dispatch.

Robotlegs Bundles


A selection of built-in bundles.



Bundles Vs Extensions


An extension integrates a single utility or library into a Robotlegs context.


A bundle bundles up a selection of extensions and configurations into a single drop-in package.





Creating a Bundle


A bundle implements the IBundle interface. When included into a context, that context is immediately passed through to the extend() method.


package com.organisation.project.bundles
{
  public class SuperDuperBundle implements IBundle
  {
    public function extend(context:IContext):void
    {
      context.install(
        SuperDuperExtensionA,
        SuperDuperExtensionB,
        SuperDuperExtensionC);
    }
  }
}



NOTE: The context instance passed to extend() may not be fully initialized.


A bundle should do little more than include required extensions.

MVCS Bundle (Classic)


This bundle installs a number of extensions and configurations for developers who are comfortable with the typical Robotlegs MVCS setup.



Included Extensions



		TraceLoggingExtension - sets up a simple trace log target


		VigilanceExtension - throws errors when warnings are logged


		InjectableLoggerExtension - allows you to inject loggers into clients


		ContextViewExtension - consumes a display object container as the contextView


		EventDispatcherExtension - makes a shared event dispatcher available


		ModularityExtension - allows the context to expose and/or inherit dependencies


		DirectCommandMapExtension - allows you to execute commands directly and detain and release command instances


		EventCommandMapExtension - an event driven command map


		LocalEventMapExtension - automatically cleans up listeners for its clients


		ViewManagerExtension - allows you to add multiple containers as “view roots”


		StageObserverExtension - watches the stage for view components using magic


		MediatorMapExtension - configures and creates mediators for view components


		ViewProcessorMapExtension - allows direct view processing (e.g. direct injection)


		StageCrawlerExtension - scans the stage at initialization for existing views


		StageSyncExtension - automatically initializes the context when the contextView lands on stage





Note: For more information on these extensions please see the extensions package.





Included Base Classes



		Command - optional, abstract command


		Mediator - default mediator implementation








Included Configs



		ContextViewListenerConfig - adds the contextView to the viewManager

Lifecycle


A lifecycle provides initialize, suspend, resume and destroy methods.


A lifecycle usually refers to a specific instance. You can provide this instance when creating your lifecycle and refer to it via the read-only target property:


const lifecycle:Lifecycle = new Lifecycle(someManagedObject);
lifecycle.target // returns someManagedObject;



For most developers, the only lifecycle you are interested in will be the context’s lifecycle, which is created by the context itself.



Overview


A lifecycle can be in any of the following settled states:



		UNINITIALIZED


		ACTIVE


		SUSPENDED


		DESTROYED





During a transition a lifecycle can be in one of the following transitionary states:



		INITIALIZING


		SUSPENDING


		RESUMING


		DESTROYING





Before initialization the lifecycle is UNINITIALIZED. From here, only initialize() is a valid transition.


During initialization the lifecycle is INITIALIZING.


Once initialized the lifecycle becomes ACTIVE.


From ACTIVE a lifecycle can be suspended or destroyed.


From SUSPENDED a lifecycle can be resumed (returning it to ACTIVE), or destroyed.


Once DESTROYED there is no way back.


Invalid transitions can be captured by listening for the LifecycleEvent.ERROR event. If no listener is attached an Error will be thrown.





Hooking in to transitions


The Lifecycle provides 4 distinct hooks: before, when and after any transition, and a callback, passed to the transition function itself, which runs between the when and after hooks.


For clarity, the ordering is:



		Before transitioning handlers run


		If there are no errors, the state is changed. If there are errors the callback passed to the transition is run, and the errors are passed to it, and we go no further


		When transitioning handlers run


		The callback passed to the transition is run


		After transitioning handlers run





A complete list of the process timing hooks:


beforeInitializing
whenIntitializing
afterIntializing

beforeSuspending
whenSuspending
afterSuspending

beforeResuming
whenResuming
afterResuming

beforeDestroying
whenDestroying
afterDestroying



Timing hooks can be chained, as they return the same interface that you use to access them.



Wait, do we really need 4 different hooks?


Each of the 4 hooks provides a particular way to hook into the transition.



		before hooks can block the transition from happening.


		when hooks are non-blocking, and run after the state has changed, but before the callback passed to the transition is run.


		after hooks are non-blocking, and run after the callback passed to the transition is run.





The callback passed to the transition can process the errors from the before handlers, and runs whether or not the transition is made. The when and after hooks only run if the transition is successful (i.e. there are no before handler errors).





Suspend and destroy handlers run backwards


The initialize and resume transitions run their handlers in the order in which they were added. The suspend and destroy transitions run their handlers in reverse, so the last handler added to a particular phase, e.g. whenDestroying, is run first during that phase.





Before, when and after hooks persist, callbacks are one-time


A handler added to beforeSuspending will be run every time the lifecycle is suspended. A callback passed to suspend() will run once only, unless you pass the same callback to the suspend() function when you run it again at a later time.





Before handlers


A lifecycle provides 4 before hooks: beforeInitializing, beforeSuspending, beforeResuming and beforeDestroying.


Handlers added to these transitions are executed when the transition starts and before any events are dispatched.


A before handler must have one of the following signatures:



		handler():void


		handler(phase:String):void


		handler(phase:String, callback:Function):void





For beforeInitializing() the phase will be preInitialize and so on.



A before handler can be asynchronous and can block the transition


If a handler accepts a callback and calls the callback with an error, the transition will be terminated, and the state will be reverted to the pre-transition state.


For more background on async handlers in Robotlegs 2 see:



		core.async.readme


		core.messaging.readme










When and After handlers


When and After handlers are executed synchronously and must have one of the following signatures:



		handler():void


		handler(phase:String):void





For whenInitializing() the phase will be initialize and so on.


When and After handlers are not passed callbacks.







Lifecycle Events


A lifecycle dispatches the following events:



		PRE_INITIALIZE


		INITIALIZE


		POST_INITIALIZE


		PRE_SUSPEND


		SUSPEND


		POST_SUSPEND


		PRE_RESUME


		RESUME


		POST_RESUME


		PRE_DESTROY


		DESTROY


		POST_DESTROY


		ERROR








Dealing with errors


There are two situations in which errors can occur



		When an invalid transition is attempted


		When a before handler calls back with an error





In both cases the lifecycle will dispatch LifecycleEvent.ERROR if a listener for that event has been attached, otherwise an error will be thrown.


When attempting a transition into a given state, a user callback may be supplied. If an error occurs, and a listener has been attached as explained above, the error will be supplied to the callback.





Handlers cannot be removed


A lifecycle manages the validity of transitions - so both initialize and destroy are, via the lifecycle’s internal state machine, one-time-only transitions. The only repeatable transitions are suspend and resume. If you need to ‘unhook’ from these transitions, we recommend you decouple your handlers and use a flag to exit-early from your handlers if the object they would deal with has been cleaned up.


private var _managedExtension:SomeExtension;

public function set managedExtension(value:SomeExtension):void
{
    _managedExtension = value;
}

private function deactivateExtension():void
{
    if (!_managedExtension) return;

    // code that actually does stuff here
}

private function activateExtension():void
{
    if (!_managedExtension) return;

    // code that actually does stuff here
}

private function addContextLifecycleHooks():void
{
    context
        .beforeSuspending(deactivateExtension)
        .beforeResuming(activateExtension)
}



If the whole Lifecycle is no longer required, just null it out, and any handlers will be released for garbage collection.





Errors are, by design, non-recoverable


The purpose of error reporting in the lifecycle is to debug during development, not to recover run-time problems. When you make use of the lifecycle, your handlers should be returning errors only on the basis of non-mutable properties: problems that were cemented at compile time, not problems that can occur only at runtime such as network availability issues, order of operations, data entry screw ups and so on.


In the case of an error in your initialize() process, the next step would be to study that error and make relevant changes in your code base, and then recompile. For this reason it’s critical that extension developers write very helpful error messages.



Good reasons for a lifecycle handler to return an error:



		The correct range of fonts wasn’t embedded


		An extension on which this extension depends hasn’t been installed


		The code is running in a sandbox that doesn’t have the permissions this extension requires


		An essential configuration object hasn’t been created (e.g. developer credentials for a licensed extension)


		Couldn’t load a module containing critical extension dependencies - e.g. an engine








Poor reasons for a lifecycle handler to return an error:



		Lost contact with the network


		User doesn’t have an account


		Couldn’t load a module containing non-critical extension dependencies - e.g. a skin










What kind of object might have a lifecycle?


Most developers will only encounter the lifecycle attached to the context.


Remember - lifecycle errors are non-recoverable. Their purpose is to provide detailed and meaningful explosions during development in order to ensure that the developer has included all the dependencies and configuration requirements of this extension.


Normal run-time errors should be dealt with in the usual ways, not through lifecycles. It is highly unlikely that providing a service with a lifecycle would be a good approach.


An extension framework - for example an entity system - might have its own lifecycle to allow other extensions to hook in to its lifecycle phases (particularly suspend / resume).





Basic Usage



For extensions accessing the context’s own lifecycle:


An example usage, for an imaginary extension which provides a developer console to the application.


_context.beforeInitializing( checkEventDispatcherInstalled )
        .beforeInitializing( checkEmbeddedFonts )
        .whenInitializing( setLocalDateTime )
        .whenInitializing( setLocalPaths )
        .whenSuspending( grabPauseTime )
        .afterSuspending( deactivateConsole )
        .whenResuming( calculatePauseInterval )
        .afterResuming( reactivateConsole )
        .beforeDestroying( offerConsoleDump )
        .afterDestroying( destroyConsole );






For managing your own object’s lifecycle


const awesomeExtension:AwesomeExtension = new AwesomeExtension();
const awesomeLifecycle:Lifecycle = new Lifecycle(awesomeExtension);

// you would now need to provide an opportunity for any interested parties to add their hooks

const callback:Function = function(errors:Object):void
{
    // log the errors
}

awesomeLifecycle.initialize(callback);

Logging


A lightweight logging implementation used internally by the framework.


We chose not to use a 3rd-party logging implementation to reduce dependencies on 3rd-party libraries. An interface exists that allows you to integrate easily with 3rd party log targets.



Defaults


The default log level is INFO. Many events logged by the framework and extensions are logged at the DEBUG level, which means they will be filtered out. Also, no log targets are installed by default, so even if you set the logLevel to DEBUG you might not see any output.


Note: The MVCS bundle changes these defaults. It adds a TraceLogTarget and sets the log level to DEBUG.





Log Targets


To set the log level to DEBUG and add a simple trace logging target do the following:


context.logLevel = LogLevel.DEBUG;
context.addLogTarget(new TraceLogTarget(context));



Or you can use the provided extension (already included in the MVCS bundle):


context.logLevel = LogLevel.DEBUG;
context.install(TraceLoggingExtension);



You can make your own custom log targets by implementing the ILogTarget interface.





Loggers


You can pull a fresh logger out of the context like so:


logger = context.getLogger(this);
logger.warn("I'm sorry {0}, I am far too well to come in to work today.", [boss]);



Or, instead of passing an instance, pass a class:


logger = context.getLogger(MyClass);



The problem here is that you need a reference to the context.


For an easier way to inject loggers please see the readme in the EnhancedLogging extension package.

Robotlegs Framework


Context [https://github.com/robotlegs/robotlegs-framework/blob/master/src/robotlegs/bender/framework/readme-context.md]


Logging [https://github.com/robotlegs/robotlegs-framework/blob/master/src/robotlegs/bender/framework/readme-logging.md]


Lifecycle [https://github.com/robotlegs/robotlegs-framework/blob/master/src/robotlegs/bender/framework/readme-lifecycle.md]


Guards [https://github.com/robotlegs/robotlegs-framework/blob/master/src/robotlegs/bender/framework/readme-guards.md]


Hooks [https://github.com/robotlegs/robotlegs-framework/blob/master/src/robotlegs/bender/framework/readme-hooks.md]

Async


To deal with asynchronous processes in the framework we must establish some conventions.





Handlers and Callbacks


Handlers and callbacks are conventions used by low level components such as the Message Dispatcher.



Handlers


A handler must have one of the following signatures:


function handler():void;
function handler(message:Object):void;
function handler(message:Object, callback:Function);



Note: the arguments can be typed where and when it makes sense to do so.


The first two forms allow for basic observing. The third form allows for asynchronous interaction.


By consuming a callback a handler is given the opportunity to suspend or terminate a process:


function handler(message:Object, callback:Function):void {
  setTimeout(callback, 100);
}



A handler can terminate a process by sending an error to the callback:


function handler(message:Object, callback:Function):void {
  if (!message) {
    callback(new Error("There was no message. Something went very wrong."));
  } else {
    callback();
  }
}



Note: A handler must eventually call the callback or the processes will never complete.


A handler should only call the callback once. The following is naughty:


function handler(message:Object, callback:Function):void {
  if (!message) {
    callback(new Error("There was no message. Something went very wrong."));
  }
  callback();
}



The code above will call the callback twice when the message is falsey. Libraries will usually guard against this sort of thing, but it’s best not to tempt fate in this case.





Callbacks


A callback must have one of the following signatures:


function callback():void;
function callback(error:Object):void;
function callback(error:Object, message:Object):void;



Note: the arguments can be typed where and when it makes sense to do so.


A callback is usually supplied to some asynchronous method:


loadUser("borris", function():void {
  trace("I think we have a borris.");
});



A callback should deal with errors:


loadUser("borris", function(error:Error):void {
  if (error) {
    throw error;
  } else {
    trace("I'm certain we have a borris.");
  }
});



A callback that doesn’t accept an error will still be called in the case of an error:


doSomethingImportant(function():void {
  trace("I'm not sure if that worked, but I know we're done.");
});



For this reason callbacks should always deal with errors (preferably before doing anything else).





safelyCallBack()


Helper function to simplify calling user facing callbacks:


safelyCallBack(callback, error, message);



Note: the helper will not protect against null callbacks. You must do that yourself:


callback && safelyCallBack(callback, error, message);



This prevents the overhead of calling safelyCallBack() when there is no callback to call. Likewise it reduces the overhead of a null check in safelyCallBack().

Guards


Guards say “yes” or “no”. They are typically used by extensions to guard against certain actions based on environmental conditions.



Guard Forms


A guard can exist in one of three forms:



		Function


		Object


		Class






Function Guards


A function guard is expected to return a Boolean value:


function randomGuard():Boolean {
    return Math.random() < 0.5;
}






Object Guards


An object guard is expected to expose an “approve” method that returns a Boolean value:


public class HappyGuard
{
    public function approve():Boolean
    {
        return true;
    }
}






Class Guards


Instantiating a Class guard should result in an object that exposes an “approve” method which returns a Boolean value:


public class SomeGuard
{
    [Inject] public var someModel:SomeModel;

    public function approve():Boolean
    {
        return someModel.enabled;
    }
}

Hooks


Hooks run before or after certain extension actions. They are typically used by extensions to run custom actions based on environmental conditions.



Hook Forms


A hook can exist in one of three forms:



		Function


		Object


		Class






Function Hooks


A function hook is simply invoked when the time is right:


function randomHook():void {
    trace("hooked!");
}






Object Hooks


An object hook must expose a “hook” method:


public class SomeHook
{
    public function hook():void
    {
        trace("hooked!");
    }
}



The “hook” method will be called each time the hook runs. The object will not be injected into.





Class Hooks


A class reference can be used as a hook:


public class SomeOtherHook
{
    [Inject] public var someModel:SomeModel;

    public function hook():void
    {
        someModel.enabled = false;
    }
}



A new object will be constructed each time the hook is run. The object will be injected into.

Robotlegs



		Home [https://github.com/robotlegs/robotlegs-framework]


		Documentation [https://github.com/robotlegs/robotlegs-framework/tree/master/src/robotlegs/bender]









          

      

      

    


    
        © Copyright .
      Created using Sphinx 1.3.1.
    

  

src/robotlegs/readme.html


    
      Navigation


      
        		
          index


        		robotlegs-framework stable documentation »

 
      


    


    
      
          
            
  
Robotlegs Bender


For now we have chosen to call this release “Bender”. See:


https://github.com/robotlegs/robotlegs-framework/issues/44



		Home [https://github.com/robotlegs/robotlegs-framework]


		Documentation [https://github.com/robotlegs/robotlegs-framework/tree/master/src/robotlegs/bender]

Robotlegs Framework Changelog:



Robotlegs 2



v2.2.0


Updates Swiftsuspenders to v2.0.0 release


Fixes #161 - Mediator synchronisation


Adds metadata check to VigilanceExtension





v2.1.0


EventMap - opens up routeEventToListener


Fixes RobotlegsInjector.createChild() bug





v2.0.0


A brand new day.





v2.0.0b8


Breaking change: Introduces IInjector


Fix: Change your injector references from Injector to IInjector





v2.0.0b7


Adds ViewProcessorMap to MVCS bundle


Adds context Lifecycle Event relaying to the EventDispatcher Extension


Adds Module Connector - for easy event relaying between modules


Removes ScopedEventDispatcherExtension - replaced by Module Connector


Removes MessageDispatcher related extensions





v2.0.0b6


Ensures that certain extensions can only be installed before context initialization.


Removes Hamcrest dependency from framework. Replaced with custom IMatcher.


Complete Command Center revamp.


Introduces DirectCommandMap (experimental).





v2.0.0b5


Fixes #109 (mediator autoRemove)


Fixes #110 (Event Command Map bug)


Fixes #112 (Modularity issues)


Fixes #113 (naughty PreDestroy)


Detain and Release events dispatched by context (Issue #104)


Switches back to official Swiftsuspenders build (v2.0.0rc2)





v2.0.0b4


Modularity timing fix.


Introduces StageCrawler Extension. This extension scans for view components that are already on stage when the context initializes.





v2.0.0b3


Lifecycle


Important: The lifecycle getter has been removed from the Context. Instead, the Lifecycle API methods have been added to the Context directly.


So, context.lifecycle.whenInitializing(callback) is now simply context.whenInitializing(callback)


Adds Vigilance Extension to MVCS Bundle


MVCS Bundle sets the LogLevel to DEBUG


Lifecycle dispatches stateChange event


Adds uninitialized getter to ILifecycle


Adds addChild() and removeChild() to IContext





v2.0.0b2


Asdocs


EventCommandMap optimisations





v2.0.0b1


Robotlegs 2 public beta







Robotlegs 1



v1.5.2


Fixed: https://github.com/robotlegs/robotlegs-framework/issues/25





v1.5.1


Fixed: https://github.com/robotlegs/robotlegs-framework/issues/24





v1.5.0


CommandMap Abstract Event Injection


For event-triggered commands the event is now also mapped to “Event”. For example:


[Inject] public var abstractEvent:Event;


[Inject] public var concreteEvent:SomeEvent;


MVCS Mediator


Added syntactic sugar methods removeViewListener() and removeContextListener()


SwiftSuspenders


Updated SwiftSuspenders to v1.6.0





v1.4.0


ViewMap & MediatorMap


contextView stage listener optimizations.


MediatorMap


Added IMediatorMap#hasMapping(viewClassOrName:*):Boolean;





v1.3.0


ApplicationDomain


Added applicationDomain getter/setter to IInjector to help with Application Domains.





v1.2.0


MediatorMap.mapType


The injectViewAs parameter is changed from expecting a Class to expecting a Class or an Array of Classes.





v1.1.2


ASDocs


Build script updated to bundle ASDocs into SWC for inline display in Flash Builder 4. Build needs to be run against Flex SDK 4.x





v1.1.1


Mediator


Fixed: http://github.com/robotlegs/robotlegs-framework/issues/#issue/6





v1.1.0


SwiftSuspenders


Updated SwiftSuspenders to v1.5.1


PLEASE NOTE: mapValue no longer injects into the value instance - the old behaviour was incorrect.


Injector


Added IInjector#hasMapping(clazz:Class, named:String = “”):Boolean;


Added IInjector#getInstance(clazz:Class, named:String = “”):*;


Added IInjector#createChild():IInjector;


CommandMap


Added ICommandMap#execute(commandClass:Class, payload:Object = null, payloadClass:Class = null, named:String = ‘’):void


Added ICommandMap#unmapEvents() - unmaps all event mappings


Added ICommandMap#detain() and release() - enables Async Commands


Mediator


Added mvcs.Mediator EventMap Sugar: addViewListener() and addContextListener()


Misc


mvcs.Context: CommandMap and MediatorMap are handed child injectors - to enable non-destructive temporary mappings.


Deprecated IContextProvider





v1.0.3


Fixed: http://github.com/robotlegs/robotlegs-framework/issues#issue/2





v1.0.2


Fixed: http://github.com/robotlegs/robotlegs-framework/issues#issue/2





v1.0.1


Updated SwiftSuspenders to v1.0.1





v1.0.0


Whammo, and the Robot has Legs. We managed to avoid hitting double digits for the RCs!





v1.0RC9


Fixes to SwiftSuspenders





v1.0RC8


Added IMediatorMap.unmapType()





v1.0RC7


ViewMap.mapClass() changed to ViewMap.mapType()





v1.0RC6


Internal changes: ContextBase cleaned up - initialize() removed. Overriding the default apparatus:


public function MyContext()
{
    injector = new SwiftSuspendersInjector(xmlConfig);
    super();
}






v1.0RC4/5


Fixes to SwiftSuspenders





v1.0RC3


Removed nometa package





v1.0RC2


CommandMap Bug fix





v1.0RC1


No changes





v0.9.8 - Untitled4


Removed ICommand


MediatorMap


Merged MediatorMap#mapModule into MediatorMap#mapType


New view mapping signature


mapType(viewClassOrName:*, mediatorClass:Class, injectViewAs:Class = null, autoCreate:Boolean = true, autoRemove:Boolean = true):void





v0.9.7 - XtensibleMixdownLoafers


Enabled XML configuration of injection points





v0.9.6 - PanelBeaten


Added the dispatch() helper method back to mvcs actors


Removed named injection points


MediatorMap


Enabled automatic creation of mediator for contextView if mapped


ContextBase


Made all ContextBase constructor arguments optional to enable declarative (mxml) Context instantiation


Nometa


Introduced Nometa implementation


Actor


Unified Model and Service into Actor
Removed Model
Removed Service


EventMap


Added strong event mapping to EventMap


CommandMap


Re-ordered the mapping arguments


From: mapEvent(commandClass:Class, eventType:String, eventClass:Class = null, oneshot:Boolean = false):void


To: mapEvent(eventType:String, commandClass:Class, eventClass:Class = null, oneshot:Boolean = false):void





v0.9.5 - BigMistake2


Removed dispatchEvent() helper method - Again, really sorry about that little mixup there


Mediator


addEventListenerTo() becomes eventMap.mapListener()





v0.9.4 - WhatProxyWhere?


Proxy renamed to Model





v0.9.3 - Untitled3


CommandMap bugfix





v0.9.2 - Untitled2


No changes!





v0.9.1 - BigMistake1


dispatch() helper method renamed to dispatchEvent()





v0.9 - ElasticChaos


Removed “as3commons-logging”:http://www.as3commons.org/


Removed EventBroadcaster


MediatorFactory renamed to MediatorMap


CommandFactory renamed to CommandMap


IInjector.bind* renamed to IInjector.map*


ICommandMap


CommandMap now accepts optional Event class parameter for stronger mapping.


The argument order had to be re-arranged so that the optional eventClass would come after the mandatory commandClass.


From: mapEvent(type:String, commandClass:Class, oneshot:Boolean = false):void;


To: mapEvent(commandClass:Class, eventType:String, eventClass:Class = null, oneshot:Boolean = false):void;


Constructor Injection


Changed all automated instantiation to use IInjector#instantiate to enable constructor injection, changed DI adapters accordingly and added new SwiftSuspenders support constructor injection


Bonus Adapters Removed


Removed Spring Action Script and SmartyPants-IOC adapters (they can be installed separately)





v0.8.1 - SwiftyPants


Added adapters for “SwiftSuspenders”:http://github.com/tschneidereit/SwiftSuspenders





v0.8 - ByeByeFlex


FlexMediator decoupled from Flex and merged into Mediator


Removed FlexMediator





v0.7 - ShortWave


Added dispatch() helper method





v0.6 - Untitled1


Introduced “as3commons-logging”:http://www.as3commons.org/





v0.5.2 - TheRegister


IMediator.onRegisterComplete renamed to IMediator.onRegister





v0.4 - OrganDonor


net.boyblack.robotlegs.* renamed to org.robotlegs.*





v0.3 - ElastoBoot


Added Spring ActionScript adapters


Provided adapters: “SmartyPants-IOC”:http://smartypants.expantra.net/


Provided adapters: “SpringActionScript”:http://www.springactionscript.org/





v0.2 - ReversiblePants


Introduced DI and reflection adapters


Provided adapters: “SmartyPants-IOC”:http://smartypants.expantra.net/





v0.1


Proof-of-concept prototype

Robotlegs Stress Testing


todo: intro





Thoughts on Performance Testing


todo: talk about benchmarking against in-test [BeforeClass] measurable player features in order to have relative speed tests that can fail based on falling out of range.



Benchmarking in Unit Tests

Robotlegs Overview



Framework


The Robotlegs framework:



		Context


		Configuration manager


		Extension installer


		Guards and hooks


		Lifecycle


		Logging





Framework [https://github.com/robotlegs/robotlegs-framework/tree/master/src/robotlegs/bender/framework]





Extensions


Robotlegs provides a number of built-in extensions:



		Command map


		Mediator map


		View manager


		And many more





Extensions [https://github.com/robotlegs/robotlegs-framework/tree/master/src/robotlegs/bender/extensions]





Bundles


Extensions are combined into convenient drop-in bundles:



		MVCS Bundle


		And others





Bundles [https://github.com/robotlegs/robotlegs-framework/tree/master/src/robotlegs/bender/bundles]





MXML


Apache Flex integration files.


MXML [https://github.com/robotlegs/robotlegs-framework/tree/master/src/robotlegs/bender/mxml]

Robotlegs Extensions


A selection of built-in extensions.



Extensions Vs Bundles


An extension integrates a single utility or library into a Robotlegs context.


A bundle bundles up a selection of extensions and configurations into a single package.







Writing An Extension


A Robotlegs extension should contain at least four things: an API, an implementation, an integration file and a readme.


The API and implementation files define the extension or utility itself.


The integration file defines how and when the extension is loaded into (and removed from) a Robotlegs context.


The readme file is the most important. It’s the first thing that you write when you sit down to build an extension. It outlines your goals, the purpose of the extension and how the extension can be used:


http://tom.preston-werner.com/2010/08/23/readme-driven-development.html


If you can’t sum up what the extension is and how it can be used in a readme file, then perhaps the scope of your extension is too big. In that case consider breaking it out into smaller, more focused extensions.



The Extension/Integration Class (IExtension)


An extension implements the IExtension interface. When included into a context, that context is immediately passed through to the extend() method:


package robotlegs.extensions.superDuper
{
  public class SuperDuperExtension implements IExtension
  {
    public function extend(context:IContext):void
    {
      trace(this, " is being installed into ", context);
      // BEWARE: the context may not be fully initialized.
    }
  }
}



NOTE: The context instance provided to configureContext() may not be fully initialized.



A Simple Extension


An extension might simply map a singleton:


public function extend(context:IContext):void
{
  context.injector.map(ISuperDuper).toSingleton(SuperDuper);
}






Synchronizing With The Context


An extension can hook into various context lifecycle phases by adding state handlers to that context when the extension is installed:


public function extend(context:IContext):void
{
  if (context.initialized)
    throw new Error("This extension must be installed prior to context initialization");

  context.beforeInitializing(beforeInitializing);
  context.afterInitializing(afterInitializing);
}

private function beforeInitializing(phase:String, callback:Function):void
{
  // `before` handlers can accept a callback as the second argument
  // note: you *must* eventually call the callback or you will stall initialization
  trace("Doing some things before the context self initializes...");
  setTimeout(callback, 1000);
}

private function afterInitializing():void
{
  // `after` handlers are synchronous and are not provided with callbacks
  trace("Doing some things now that the context is initialized...");
}



For more information on message handling and managed objects see:



		framework/readme-async


		framework/readme-message-dispatcher


		framework/readme-lifecycle










Packaging A Robotlegs-Specific Extension


The source for an extension should be packaged thusly:


src
  robotlegs
    extensions
      superDuper
        readme.md
        SuperDuperExtension (implements IContextConfig)
        api
          ISuperDuper
        impl
          SuperDuper



We can clearly spot the API, implementation and integration classes above.


The api package should include any classes or interfaces that the typical user would come into contact with. The impl package contains classes that the typical user would not import.





Unit Tests


The unit tests for an extension should be packaged thusly:


test
  robotlegs
    extensions
      superDuper
        SuperDuperExtensionTest
        SuperDuperExtensionTestSuite
            api
              SuperDuperErrorTest
        impl
          SuperDuperTest
        support
          SuperDuperFrequencyModulator



The ExtensionTest tests the extension integration class itself. The implementation package contains tests for implementation classes. And the TestSuite catalogs all of the test cases.







Distributing an Extension


todo: recommend GitHub. Recommend unit tests.


robotlegs-extensions-SuperDuper





Dev Notes

Matching Extension



Overview


TypeMatchers allow you to build up rich descriptions of objects based on their type - the class, superclass chain and interfaces the implement.





Extension Installation


This extension does not need to be installed, it is used by the MediatorMap extension, but can also be used in other extensions.





TypeMatcher Usage


You create a typeMatcher using new:


new TypeMatcher()



Any number of calls to the following api can be chained:


allOf(...types)
anyOf(...types)
noneOf(...types)



The parameter ...types can be an array of types or simply a list. Using an array will allow you to use sets of types to configure multiple matchers.


TypeMatchers are locked as soon as they have been used for matching, but you can explicitly lock your TypeMatcher with a call to


lock()



Once locked, the only api available is


clone()



This creates a duplicate TypeMatcher, which is unlocked, so that you can further customise this matcher.


TypeMatchers are not internally checked for conflicts, so it’s your responsibility to ensure that your rules make sense.


There is no ‘or’ provided. Instead, use multiple matchers and map each of them to the same rules - ensuring that your rules aren’t nested, or multiple matches will occur. For guidance on testing your matchers check out the TypeMatcher unit tests.





PackageMatcher Usage


The package matcher has the following api:


require(fullPackage:String)
anyOf(...packages)
allOf(...pacakges)



You can only require one package. Multiple calls to anyOf and allOf can be chained. Multiple calls to require will throw an error.


The package matcher will be locked the first time it is used, but can be explicitly locked with a call to


lock();






Building your own Matchers


... info to follow

Robotlegs


Robotlegs is an ActionScript application framework for Flash and Flex. It offers:



		Dependency injection


		Module management


		Command management


		View management


		Plug-and-play extensions






Download


http://www.robotlegs.org/





Documentation & Support


The framework documentation exists as README files in the repository. The best way to read them is through GitHub:


Documentation [https://github.com/robotlegs/robotlegs-framework/tree/master/src/robotlegs/bender]


Support [http://knowledge.robotlegs.org/]



Developed with pleasure using IntelliJ IDEA


[image: ]







Robotlegs 2 (What’s New?)



		The fluent API makes your code more readable


		An extension mechanism makes the framework highly customizable


		Bundles allow you to get applications started quickly


		Lightweight, built-in logging shows you what the framework is doing


		Mediators can be mapped to abstract types and interfaces


		View processing has been significantly optimized


		A context can process more than one display list (PopUp support)


		Greatly simplified module (multi-context) support





Change log [https://github.com/robotlegs/robotlegs-framework/blob/master/changelog.md]







Quickstart



Creating A Context


To create a Robotlegs application or module you need to instantiate a Context. A context won’t do much without some configuration.


Plain ActionScript:


_context = new Context()
    .install(MVCSBundle)
    .configure(MyAppConfig, SomeOtherConfig)
    .configure(new ContextView(this));



We install the MVCSBundle, which in turn installs a number of commonly used Extensions. We then add some custom application configurations.


We pass the instance “this” through as the “contextView” which is required by many of the view related extensions. It must be installed after the bundle or it won’t be processed. Also, it should always be added as the final configuration as it may trigger context initialization.


Note: You must hold on to the context instance or it will be garbage collected.


Flex:


<fx:Declarations>
    <rl2:ContextBuilder>
        <mvcs:MVCSBundle/>
        <config:MyAppConfig/>
    </rl2:ContextBuilder>
</fx:Declarations>



Note: In Flex we don’t need to manually provide a “contextView” as the builder can determine this automatically.


Framework [https://github.com/robotlegs/robotlegs-framework/tree/master/src/robotlegs/bender/framework]





Context Initialization


If a ContextView is provided the Context is automatically initialized when the supplied view lands on stage. Be sure to install the ContextView last, as it may trigger context initialization.


If a ContextView is not supplied then the Context must be manually initialized.


_context = new Context()
    .install(MyCompanyBundle)
    .configure(MyAppConfig, SomeOtherConfig)
    .initialize();



Note: This does not apply to Flex MXML configuration as the ContextView is automatically determined and initialization will be automatic.


ContextView [https://github.com/robotlegs/robotlegs-framework/tree/master/src/robotlegs/bender/extensions/contextView]





Application & Module Configuration


A simple application configuration file might look something like this:


public class MyAppConfig implements IConfig
{
    [Inject]
    public var injector:IInjector;

    [Inject]
    public var mediatorMap:IMediatorMap;

    [Inject]
    public var commandMap:IEventCommandMap;

    [Inject]
    public var contextView:ContextView;

    public function configure():void
    {
        // Map UserModel as a context enforced singleton
        injector.map(UserModel).asSingleton();

        // Create a UserProfileMediator for each UserProfileView
        // that lands inside of the Context View
        mediatorMap.map(UserProfileView).toMediator(UserProfileMediator);

        // Execute UserSignInCommand when UserEvent.SIGN_IN
        // is dispatched on the context's Event Dispatcher
        commandMap.map(UserEvent.SIGN_IN).toCommand(UserSignInCommand);

        // The "view" property is a DisplayObjectContainer reference.
        // If this was a Flex application we would need to cast it
        // as an IVisualElementContainer and call addElement().
        contextView.view.addChild(new MainView());
    }
}



The configuration file above implements IConfig. An instance of this class will be created automatically when the context initializes.


We Inject the utilities that we want to configure, and add our Main View to the Context View.


Framework [https://github.com/robotlegs/robotlegs-framework/tree/master/src/robotlegs/bender/framework]



An Example Mediator


The mediator we mapped above might look like this:


public class UserProfileMediator extends Mediator
{
    [Inject]
    public var view:UserProfileView;

    override public function initialize():void
    {
        // Redispatch an event from the view to the framework
        addViewListener(UserEvent.SIGN_IN, dispatch);
    }
}



The view that caused this mediator to be created is available for Injection.


MediatorMap [https://github.com/robotlegs/robotlegs-framework/tree/master/src/robotlegs/bender/extensions/mediatorMap]





An Example Command


The command we mapped above might look like this:


public class UserSignInCommand extends Command
{
    [Inject]
    public var event:UserEvent;

    [Inject]
    public var model:UserModel;

    override public function execute():void
    {
        if (event.username == "bob")
            model.signedIn = true;
    }
}



The event that triggered this command is available for Injection.


EventCommandMap [https://github.com/robotlegs/robotlegs-framework/tree/master/src/robotlegs/bender/extensions/eventCommandMap]









Building and Running the Tests



Requirements


Whilst Robotlegs can be used for Flex 3 & 4 and plain ActionScript projects, the library must be built with the Flex 4.6 SDK or above.





Building with ANT


Copy the “user.properties.eg” file to “user.properties” and edit it to point to your local Flex SDK. Then run:


ant package





Building with Maven


See: Maven-README





Building with Buildr on OSX



		Install XCode 3 or 4





		check RubyGems version


$ gem -v
1.8.1





		update RubyGems if version is less than 1.3.6


$ sudo gem update –system





		install Bundler


$ sudo gem install bundler





		run Bundler to install dependencies


$ bundle install





		run Buildr to build RobotLegs & run Tests


$ bundle exec buildr test





		open test report


$ open reports/flexunit4/html/index.html








Example output of this process [https://gist.github.com/1336238]







Robotlegs 1


The source for Robotlegs 1 can be found in the version1 branch:


Robotlegs Version 1 Branch [https://github.com/robotlegs/robotlegs-framework/tree/version1]

Mediator Map Extension



Overview


The mediator map provides automagic mediator creation for mapped views landing on the stage, as well as a hook for other utilities to create mediators for non-view objects, triggered in whatever manner suits them.





Extension Installation


This extension is already installed in the MVCS bundle, but if you’re not using that:


context.install(MediatorMapExtension);




Default access to the map is by injecting against the IMediatorMap interface


[Inject]
public var mediatorMap:IMediatorMap;








MediatorMap Usage


The Robotlegs 2 mediatorMap is designed to support co-variant mediation. This means that multiple mediators can be mediating on behalf of one object, and that the combination of mediators created is driven by the object’s type (class, superclasses and interfaces).



Making mappings


You map either a specific type or a TypeMatcher to the mediator class you want to be created.


mediatorMap.map(SomeType).toMediator(SomeMediator);

mediatorMap.mapMatcher(new TypeMatcher().anyOf(ISpaceShip, IRocket)).toMediator(SpaceCraftMediator);



We provide a TypeMatcher and PackageMatcher. TypeMatcher has allOf, noneOf, anyOf. For more complex logic (equivalent of ‘or’) you can simply make multiple mappings. For more details on type and package matching, see:



		robotlegs.bender.extensions.matching.readme





You can optionally add guards and hooks:


map(SomeClass).toMediator(SomeMediator)
    .withGuards(NotOnTuesdays)
    .withHooks(ApplySkin, UpdateLog);



Guards and hooks can be passed as arrays or just a list of classes.


Guards will be injected with the view to be mediated, and hooks can be injected with both the view and the mediator (these injections are then cleaned up so that mediators are not generally available for injection).


For more information on guards and hooks check out:



		robotlegs.bender.framework.readme-guards


		robotlegs.bender.framework.readme-hooks








Removing mappings


mediatorMap.unmap(SomeClass).fromMediator(SomeMediator);

mediatorMap.unmapMatcher(someTypeMatcher).fromMediator(SomeMediator);

mediatorMap.unmap(SomeClass).fromMediators();






Mediating views automatically


In Robotlegs 2, stage-event listening is centralised to a ViewManager. The ViewManager listens for views landing on the stage, and being removed from stage, and informs interested parties, such as the mediatorMap, accordingly.


Assuming you’re listening to your contextView, any view that lands on the contextView can be mediated if it matches a mapping you’ve already created.





Mediating objects manually


The mediatorMap is able to mediate non-view objects. However, you’ll need to implement your own strategy for deciding when these objects should be mediated and unmediated. Map your rules as normal, and then use:


mediatorMap.mediate(item);

mediatorMap.unmediate(item);






Packages excluded from automatic mediation


For efficiency, classes from the following packages are excluded from automatic mediation:


flash
mx
spark






Flex specifics


Flex UIComponents will have their mediators ‘paused’ until creationComplete has fired.





Reparenting


When a view is reparented it fires both the remove and the added events. The mediator will be destroyed and then recreated. Any guards will be reapplied, and hooks will run again. You can guard against this situation in your hooks if it is potentially a problem (by keeping a cache of skinned views or checking some property on the view).





Unmapping does not equal unmediating


Unmapping removes the rule. It will not destroy existing mediators related to that mapping. To remove mediators as well you’ll need to unmediate the objects concerned manually.





Mapping and Unmapping is robust to redundancy



		Duplicating an existing mapping, using the same guards and hooks, will not produce an error


		Unmapping a mapping that does not exist will not produce an error


		Repeating a mapping but with different guards and hooks will produce an error.
		if you use guards or hooks that were not previously used, the error is synchronous


		if you omit guards or hooks that were previous used, the error will happen as early as possible, which is the next time the mapping is used by the mediatorMap


		if you omit guards or hooks and this mapping is never used again, you will never see the error (but it shouldn’t matter)
















Mediators



Extend, follow convention, or bake your own


Mediators should observe one of the following forms:



		Extend the base mediator class and override initialize and, if needed, destroy.
		If you override destroy, don’t forget to call super.destroy() as this is where event listening cleanup is triggered.








		Don’t extend the base mediator class, and provide functions initialize() and, if needed, also destroy().


		Don’t follow this convention, and use the [PostConstruct] metadata tag to ensure your initialization function is run
		note that this approach is not tailored for views extending Flex UIComponent, where initialization should be deferred until after creationComplete, so you will need to either provide for this in your implementation or use one of the methods above.












Example Mediators


A mediator that extends the MVCS Mediator might look like this:


public class UserProfileMediator extends Mediator
{
    [Inject]
    public var view:UserProfileView;

    override public function initialize():void
    {
        // Redispatch the event to the framework
        addViewListener(UserEvent.SIGN_IN, dispatch);
    }
}



You do not have to extend the MVCS mediator:


public class UserProfileMediator
{
    [Inject]
    public var view:UserProfileView;

    [Inject]
    public var dispatcher:IEventDispatcher;

    public function initialize():void
    {
        view.addEventListener(UserEvent.SIGN_IN, dispatcher.dispatch);
    }

    public function destroy():void
    {
        view.removeEventListener(UserEvent.SIGN_IN, dispatcher.dispatch);
    }
}



Notice that we could not use the handy “addViewListener” sugar. Also, we now need to manually clean up any listeners we have attached.







Mediator base class provides some useful functionality


Mediator base class provides the following internal API, used for managing listeners and dispatching events.


addViewListener(eventString:String, listener:Function, eventClass:Class = null):void

addContextListener(eventString:String, listener:Function, eventClass:Class = null):void

removeViewListener(eventString:String, listener:Function, eventClass:Class = null):void

removeContextListener(eventString:String, listener:Function, eventClass:Class = null):void

dispatch(event:Event):void



You can also access the injected eventMap directly, for example to listen to a subcomponent.


For more details on the local eventMap see:



		robotlegs.bender.extensions.localEventMap.readme

Event Map Extension



Overview


An Event Map allows for mapping event listeners to a specific Event Dispatcher instance and provides some convenience methods not available with ordinary Event Dispatchers.





Basic Usage


eventMap.mapListener(eventDispatcher, FooEvent.Foo, listener);



The Event Map keeps a list of listeners for easy removal.





Unmapping all listeners


eventMap.unmapListeners();



Removes all listeners registered through mapListener.





Unmapping a specific listener


eventMap.unmapListener(eventDispatcher, FooEvent.FOO, listener);

## Suspending and resuming event listening

The Event Map provides methods to suspend and resume event listening.

```as3
eventMap.mapListener(eventDispatcher, FooEvent.Foo, listener);
eventDispatcher.dispatchEvent(FooEvent.FOO); //listener is called

eventMap.suspend();
eventDispatcher.dispatchEvent(FooEvent.FOO); //listener is not called

eventMap.resume();
eventDispatcher.dispatchEvent(FooEvent.FOO); //listener is called again






Strong typing events


Optionally you can register listeners to a concrete event class. This ensures that the listener will only be called if the event is an instance of a specific event class.


//registers the listener to the concrete FooEvent class
eventMap.mapListener(eventDispatcher, FooEvent.FOO, listener, FooEvent);

//the listener will not be called, since the event is an instance of Event not FooEvent
eventDispatcher.dispatchEvent(new Event(FooEvent.FOO)); 






Extension Installation


_context = new Context()
    .install(LocalEventMapExtension);






Extension Usage


An instance of IEventDispatcher is mapped into the context during extension installation. This instance can be injected into clients.


[Inject]
public var eventDispatcher:IEventDispatcher;

Maven usage



Installation


Step 1: Install Maven 3.0.3+


Download from here [http://maven.apache.org/download.html]


Step 2: Ensure maven binaries are on your PATH (ie. you can run mvn from anywhere)


Follow the installation instructions from here [http://maven.apache.org/download.html#Installation].


Step 3: CD to the robotlegs root (where the POM.xml lives)


Step 4: Install the third party dependencies into your local repository



		In OS X


  bash -x maven-runonce.sh






		In Windows


  maven-runonce.bat










Note: we do this because unlike Ant, Maven requires all dependencies live within a repository somewhere. Because none of these dependencies are hosted externally on a remote repository, we need to install them locally into the repository. Going forward, the hope is that many of these dependencies will live in Maven Central, negating the need to install them locally. JM.






To build


In order to build Robotlegs from source, run the following command in the Robotlegs root folder.


mvn clean install -s settings.xml






Alternative to supplying settings


If you would rather not supply -s settings.xml for every build, you can add the repository information from settings.xml into your local repository. Alternatively, you can simply copy the robotlegs settings.xml to ~/.m2/settings.xml where ~ is your user directory (\Users\username in Win7+)



Why do we need the settings file? While Flexmojos lives within Maven Central, its dependencies (such as the Flex compiler) and flash project dependencies (such as framework SDKs) do NOT. They live within the flexgroup branch off (sonatype)[http://repository.sonatype.org/content/groups/flexgroup/].

View Manager Extension


TODO: write things

Modularity Extension



Overview


The modularity extensions wires contexts into a hierarchy based on the context view and allows for inter-modular communication.





Basic usage


Communication between modules is facilitated by the Module Connector.


Setup to allow sending events from one module to the other:


//ModuleAConfig.as

[Inject]
public var moduleConnector: IModuleConnector;

moduleConnector.onDefaultChannel()
    .relayEvent(WarnModuleBEvent.WARN);



Setup to allow reception of events from another module:


//ModuleBConfig.as
[Inject]
public var moduleConnector:IModuleConnector;

moduleConnector.onDefaultChannel()
    .receiveEvent(WarnModuleBEvent.WARN);



Now ModuleB can map commands to the event, or allow mediators to attach listeners to it:


eventCommandMap.map(WarnModuleBEvent.WARN)
    .toCommand(HandleWarningFromModuleACommand);



All ModuleA needs to do is dispatch the event:


eventDispatcher.dispatchEvent(new WarnModuleBEvent(WarnModuleBEvent.WARN);






Named channels


If you want to sandbox the communication between two modules, you can use named channels:


//ModuleAConfig.as
moduleConnector.onChannel('A-and-B')
    .relayEvent(WarnModuleBEvent.WARN);



//ModuleBConfig.as
moduleConnector.onChannel('A-and-B')
    .receiveEvent(WarnModuleBEvent.WARN);






Requirements


This extension requires the following extensions:



		ContextViewExtension








Extension Installation


_context = new Context()
    .install(ContextViewExtension, ModularityExtension)
    .configure(new ContextView(this));



In the example above we provide the instance “this” to use as the Context View. We assume that “this” is a valid DisplayObjectContainer.


By default the extension will be configured to inherit dependencies from parent contexts and expose dependencies to child contexts. You can change this by supplying parameters to the extension during installation:


_context = new Context()
    .install(ContextViewExtension)
    .install(new ModularityExtension(true, false))
    .configure(new ContextView(this));



The example context above inherits dependencies from parent contexts but does not expose its own dependencies to child contexts. However, child contexts may still inherit dependencies from this context’s parents.

Direct Command Map



Overview


The Direct Command Map allows you to execute commands directly and to pin and release commands in/from memory.





Command Execution



Executing a single command


directCommandMap
    .map(RetrieveFlashVarsCommand)
    .execute();






Executing a sequence of commands


directCommandMap
    .map(DoThisFirstCommand)
    .map(DoThisNextCommand)
    .map(DontForgetThisOneTooCommand)
    .map(FinallyDoThisCommand)
    .execute();



The sequence is handled synchronized, i.e. once the code in a command finishes execution the Direct Command Map immediately moves on to the execution of the next command in line.





Mapping guards and hooks


You can optionally add guards and hooks:


directCommandMap
    .map(CalculateAnswerToLifeTheUniverseAndEverythingCommand)
        .withGuards(DeepThoughtBuiltGuard)
        .withHooks(StartAnnouncementCeremonyHook)
    .map(BuildEarthCommand)
        .withGuards(WeDontKnowTheQuestionGuard)
        .withHooks(AlertTheMiceHook)
    .execute();



For more information on guards and hooks check out:



		robotlegs.bender.framework.readme-guards


		robotlegs.bender.framework.readme-hooks








Mapping a payload


You can optionally pass a payload object to the execute method, its values will be injected into the commands:


const payload : CommandPayload = new CommandPayload();
payload.addPayload(new Adieu('So long and thanks for all the fish!'), Adieu);

directCommandMap
    .map(ExtractDolphinsCommand)
    .execute(payload);



For each payload item you need to provide a value and a class to map the value against. You have to make sure that each mapped value class is unique for the payload instance, otherwise the injection mapping will be overwritten.


This WILL OVERWRITE the first String value:


const payload : CommandPayload = new CommandPayload();
payload.addPayload('So long and thanks for all the fish!', String);
payload.addPayload('Hope you enjoyed the tuna!', String); //overwrites the previous String value



Your command receives the payload values as normal injections:


[Inject]
public var adieu : Adieu;

public function execute():void
{
    log.warn(adieu.toString());
}






Note: Payload values are temporarily mapped for injection and sandboxed


All payload values are only mapped for the duration of the execution (sequence) and they won’t be injected into any other freshly created objects besides the commands, i.e. they’re sandboxed.







Command instance pinning to and releasing from memory


By concept commands are short-lived, once their execution is finished they are released for garbage collection.
However, the Direct Command Map allows commands to explicitly pin themselves into memory and release when necessary:


[Inject]
public var directCommandMap : IDirectCommandMap;

[Inject]
public var importantService : IImportantService;

public function execute():void
{
    directCommandMap.detain(this);
    importantService.doSomethingImportant(onServiceCallComplete);
}

private function onServiceCallComplete():void
{
    directCommandMap.release(this);
}








Direct Command Map extension



Requirements


This extension requires the following extension:



		CommandCenterExtension








Extension Installation


_context = new Context()
    .install(DirectCommandMapExtension);






Extension Usage


An instance of IDirectCommandMap is mapped into the context during extension installation. This instance can be injected into clients and used as below.


[Inject]
public var directCommandMap : IDirectCommandMap;

ContextView Extension (Bundled with MVCS)


The Context View Extension adds a configuration processor to the context that consumes a ContextView object and maps the provided view as a DisplayObjectContainer into the context. Many extensions require a DisplayObjectContainer to be present in order to function correctly.



Installation



During Context Construction


_context = new Context()
    .install(ContextViewExtension)
    .configure(new ContextView(this));



Note: The ContextViewExtension must be installed before the ContextView is provided or it will not be processed.


In the example above we provide the instance “this” to use as the view. We assume that “this” is a valid DisplayObjectContainer.









StageSync Extension (Bundled with MVCS)


The Stage Sync Extension waits for a ContextView to be added as a configuration, and initializes and destroys the context based on the contextView’s stage presence.


_context = new Context()
    .install(ContextViewExtension, StageSyncExtension)
    .configure(new ContextView(this));



Due to the StageSyncExtension the context above will automatically initialize when the Context View lands on the stage.



Manual Context Initialization


If you do not install the StageSync Extension or do not provide a Context View you must initialize the context manually:


_context = new Context()
    .install(MyCompanyBundle)
    .configure(MyAppConfig, SomeOtherConfig)
    .initialize();

CommandCenter


This extension is used by command based extensions such as the Event Command Map.


It is not intended to be used directly.

Event Dispatcher Extension



Overview


The event dispatcher extension simply maps a shared event dispatcher into a context. The extension is required by many event driven extensions.





Extension Installation


_context = new Context()
    .install(EventDispatcherExtension);



You can provide the dispatcher instance you wish to use manually if you so desire:


_context = new Context()
    .install(new EventDispatcherExtension(dispatcher));






Extension Usage


An instance of IEventDispatcher is mapped into the context during extension installation. This instance can be injected into clients.


[Inject]
public var eventDispatcher:IEventDispatcher;

Enhanced Logging


Provides a number of extensions related to logging.



TraceLoggerExtension


Adds a TraceLogTarget to the context. This will simply trace() log messages.





InjectableLoggerExtension


If you install the InjectableLoggerExtension (already included in the MVCS bundle), getting loggers becomes even easier - simply inject an ILogger:


public class MyClass
{
    [Inject]
    public var logger:ILogger;

    public function MyClass()
    {
        // don't try to access the logger in your constructor
        // it won't have been set yet
    }

    public function explode(size:String):void
    {
        logger.info("There was a {0} explosion!", [size]);
    }
}






InjectorActivityLoggingExtension


Ever wanted to know what the injector is up to? The InjectorActivityLoggingExtension adds listeners to the injector and logs its every move.


The injection and mappings events are logged at the DEBUG level, so be sure to set your logLevel to debug if you want to see them.


WARNING: This extension will seriously degrade performance. Use it for fun, not profit.

Vigilance Extension



Overview


The Vigilance Extension throws Errors when warnings, errors or fatal messages are logged. This keeps the good times rolling.

Robotlegs MXML Integration


Robotlegs MXML integration files.

Code metric tools



checkASDocs.rb


Does a crude check of whether each public function in each code file has a comment block immediately ahead of it.


Usage:


ruby tools/checkASDocs.rb src


Will list out functions requiring comment blocks.


ruby tools/checkASDocs.rb src true


Will list out functions which have comment blocks.


Note: No attempt is made to verify whether all parameters, return values and so on are covered. The block may not even be ASDoc compatible, and the content of the block might be gibberish.

View Processor Map Extension



Overview


The view processor map provides automagic processing of mapped views landing on the stage and manual processing where preferred.





Extension Installation


context.install( ViewProcessorMapExtension );




Default access to the map is by injecting against the IViewProcessorMap interface


[Inject]
public var viewProcessorMap:IViewProcessorMap;








ViewMap Usage


The Robotlegs 2 view map is designed to allow views to be ‘processed’, driven by the object’s type (class, superclasses and interfaces). The most typical ‘processor’ would inject the view via inspection - and this processor is built in.


You can create your own processor, for example to do property injection without inspection, skinning, localisation and so on.



Making mappings


You map either a specific type or a TypeMatcher to the class or instance of processor you want to be used.


viewProcessorMap.map(SomeType).toProcessor(FastInjector);

viewProcessorMap.mapMatcher(new TypeMatcher().anyOf(ISpaceShip, IRocket)).toProcessor(SpacecraftSkinner);

// you can also use an instance as the processor, in this case, to avoid inspection when doing property injection

viewProcessorMap.map(SomeType).toProcessor( new FastPropertyInjector( { userID:UserID, animationSettings:AnimationSettings } ) );




Shortcut method for the most common case: injection by inspection


viewProcessorMap.map(SomeType).toInjection();






Type / Package Matching


We provide a TypeMatcher and PackageMatcher. TypeMatcher has allOf, noneOf, anyOf. For more complex logic (equivalent of ‘or’) you can simply make multiple mappings. For more details on type and package matching, see:



		robotlegs.bender.extensions.matching.readme








Adding guards and hooks


You can optionally add guards and hooks:


map(SomeClass).toProcess(LocaliseText).withGuards(NotOnTuesdays).withHooks(UpdateLog);

// in the situation where you just want guards and hooks, and no processing needs to be done

map(SomeClass).toNoProcess().withGuards(NotOnTuesdays).withHooks(ApplySkin, UpdateLog);



Guards and hooks can be passed as arrays or just a list of classes.


Guards and hooks will be injected with the view (these injections are then cleaned up so that views are not generally available for injection).


The guards and hooks run prior to the process method on the processor. Processors are NOT automatically made available for injection to hooks.


In the case where you need access to the processor in a hook, pass a class rather than an instance as the process, which will then be used as a singleton and mapped for injection, and inject against the processor class (or interface) in your hook.


For more information on guards and hooks check out:



		robotlegs.bender.framework.readme-guards


		robotlegs.bender.framework.readme-hooks












Utility processors provided





ViewInjectionProcessor


Injects view by passing it to the injector, where it will be inspected and then injected. You can access this via either of the following:


map(SomeType).toInjection();
map(SomeType).toProcess(ViewInjectionProcessor);






FastPropertyInjector


Allows injection of properties (by the injector), without describing the object type. You provide names and types of injection points in the configuration object passed to the constructor.


map(ViewThatIsTooExpensiveToInspect).toProcess(new FastPropertyInjector({gravity:Gravity, bounce:Bounce}));






PropertyValueInjector


Allows injection of values directly, using only property names, so that the injector is never consulted.


map(ViewNeedingQuickParams).toProcess(new PropertyValueInjector({gravity:9.8, bounce:4}));






MediatorCreator


Allows you to use mediators via the viewProcessorMap - for example in the case where you want to minimise the size of the framework within your application.


map(ViewNeedingMediator).toProcess(new MediatorCreator(SomeMediator));



The mediator class passed should implement set viewComponent, initialize and destroy methods as required (only those provided will be called).


A difference between the mediatorMap and the viewProcessorMap: in the mediatorMap, mediators can be injected into hooks. In the viewProcessorMap they aren’t mapped for injection at all.



Creating custom processors


Processors need to implement two methods:


process(view:ISkinnable, class:Class, injector:IInjector):void;
unprocess(view:ISkinnable, class:Class, injector:IInjector):void;



These methods are checked by duck typing, rather than forcing you to implement an interface, so that you can use stricter typing on the view argument passed. The class is passed to avoid you having to re-inspect the object (in most cases the viewProcessorMap will already have obtained this information).


In many cases, unprocess will simply be an empty function, but some processes may need to ‘clean up’ - for example removing listeners and so on.


Where a processor is mapped as a class, a singleton instance will be instantiated via the injector.


In most cases, processors shouldn’t be stateful with respect to the views that they handle, other than in keeping a (weak) cache of which objects have already been processed, both for the purposes of unprocessing and for not re-processing the same object.


The injector passed is an instance of the local injector for the map. If your processor doesn’t need access to the injector and you want to avoid importing it, simply type this 3rd argument Object or *.


You may wish to use a childInjector for local mappings within the process.





Removing mappings


viewProcessorMap.unmap(SomeClass).fromProcess(FastInject);

viewProcessorMap.unmapMatcher(someTypeMatcher).fromProcess(processorInstance);

viewProcessorMap.unmap(SomeClass).fromProcesses();






Processing views automatically


In Robotlegs 2, stage-event listening is centralised to a ViewManager. The ViewManager listens for views landing on the stage, and being removed from stage, and informs interested parties, such as the viewProcessorMap, accordingly.


Assuming you’re listening to your contextView, any view that lands on the contextView can be processed if it matches a mapping you’ve already created.





Processing objects manually


If you don’t want the overhead of listening for views landing on the stage, you’ll need to implement your own strategy for deciding when views should be processed and unprocessed. Map your rules as normal, and then use:


viewProcessorMap.process(item);

viewProcessorMap.unprocess(item);






Packages excluded from automatic processing


For efficiency, classes from the following packages are excluded from automatic processing:


flash
mx
spark






Flex specifics


Flex UIComponents can have their processors ‘paused’ until creationComplete has fired - this is something you would implement in your custom processor.





Reparenting


When a view is reparented it fires both the remove and the added events. On remove, the view will be unprocessed. On add it will be offered to be reprocessed (if guards allow it). The default processor (the normal injector) will keep a cache of processed views (as keys in a weak Dictionary to avoid holding references that should be released), and will not reprocess them. However, guards and hooks will potentially be re-run - you can employ your own tracking/caching strategy to avoid repeating hooks that don’t need to re-run.


Your own processor can employ which ever strategy makes sense to you.





Unmapping does not equal unprocessing


Unmapping removes the rule. It will not unprocess views related to that mapping. If necessary, you’ll can unprocess the views concerned manually.





Mapping and Unmapping is robust to redundancy



		Duplicating an existing mapping, using the same guards and hooks, will not produce an error


		Unmapping a mapping that does not exist will not produce an error


		Repeating a mapping but with different guards and hooks will produce an error.
		if you use guards or hooks that were not previously used, the error is synchronous


		if you omit guards or hooks that were previous used, the error will happen as early as possible, which is the next time the mapping is used by the mediatorMap


		if you omit guards or hooks and this mapping is never used again, you will never see the error (but it shouldn’t matter)

Event Command Map



Overview


An Event Command Map executes commands in response to events on a given Event Dispatcher.





Basic Usage


eventCommandMap
    .map(SignOutEvent.SIGN_OUT)
    .toCommand(SignOutCommand);

eventDispatcher.dispatchEvent(new SignOutEvent(SignOutEvent.SIGN_OUT));






Strong typing events


Optionally you can map commands to a concrete event class. This ensures that the command will only be executed if the event instance is of a specific event class.


eventCommandMap
    .map(FooEvent.FOO, FooEvent) //mapping the command to the concrete FooEvent class
    .toCommand(FooCommand);

//the command will NOT be executed, since the event instance is of type Event
eventDispatcher.dispatchEvent(new Event(FooEvent.FOO)); 






Event instance access inside commands


Commands are automatically injected with the event instance that triggered them, resolved to the type used for the mapping.



Concrete event types


eventCommandMap
    .map(SignOutEvent.SIGN_OUT, SignOutEvent) //the second parameter defines the type of event instance injection
    .toCommand(SignOutCommand);



//SignOutCommand
[Inject]
public var event: SignOutEvent; //as mapped

public function execute():void{
    //do something useful with event
}






Abstract event types


eventCommandMap
    .map(SignOutEvent.SIGN_OUT, Event) //the second parameter defines the type of event instance injection
    .toCommand(SignOutCommand);



//SignOutCommand
[Inject]
public var event: Event; //as mapped

public function execute():void{
    //do something useful with event
}






Note: this deviates from Robotlegs v1 functionality


In Robotlegs v1 an event instance was automatically mapped both to the concrete event type and the abstract Event type.







Mapping ‘once’ commands


If you know that you only want your command to fire once, and then be automatically unmapped:


eventCommandMap
    .map(SignOutEvent.SIGN_OUT, SignOutEvent)
    .toCommand(SignOutCommand)
    .once();






Mapping guards and hooks


You can optionally add guards and hooks:


eventCommandMap
    .map(SignOutEvent.SIGN_OUT, SignOutEvent)
    .toCommand(SignOutCommand)
    .withGuards(NotOnTuesdays)
    .withHooks(UpdateLog);



Guards and hooks can be passed as lists of classes, objects or function references.


Guards will be injected with the event that fired, and hooks can be injected with both the event and the command (these injections are then cleaned up so that events and commands are not generally available for injection).


For more information on guards and hooks check out:



		robotlegs.bender.framework.readme-guards


		robotlegs.bender.framework.readme-hooks








Note: strictly one mapping per-event-per-command


You can only make one mapping per event-command pair. You should do your complete mapping in one chain.


So - the following will issue a warning:


eventCommandMap.map(SomeEvent.STARTED).toCommand(SomeCommand);
eventCommandMap.map(SomeEvent.STARTED).toCommand(SomeCommand); // warning



If you intend to change a mapping you should unmap it first.







Event Command Map Extension



Requirements


This extension requires the following extensions:



		EventDispatcherExtension








Extension Installation


_context = new Context().install(
    EventDispatcherExtension,
    EventCommandMapExtension);



Or, assuming that the EventDispatcherExtension has already been installed:


_context.install(EventCommandMapExtension);






Extension Usage


An instance of IEventCommandMap is mapped into the context during extension installation. This instance can be injected into clients and used as below.


[Inject]
public var eventCommandMap:IEventCommandMap;

